



# Taux de variation instantané et interprétation graphique

© Pierre Lantagne

Enseignant retraité du Collège de Maisonneuve

La première version de ce document est parue sous la version Maple 6. Ce document est la suite du document « Taux de variation moyen et interprétation graphique ». La macro-commande `textplot` de l'extension `plots` y a été introduite et nous l'utiliserons pour bien documenter les graphiques.

Bonne lecture à tous !

\* Ce document Maple est exécutable avec la version 2020.1

## Initialisation

```

> restart;
> with(plots,display,textplot,setoptions):
  setoptions(size=[400,400],axesfont=[times,roman,8],color=navy):

```

## Taux de variation instantané

Saisissons à nouveau la fonction TVM créée dans le document « Taux de variation moyen et interprétation graphique ». Les points P et Q sont les points d'une fonction f.

```

> TVM:=(P,Q)->( Q[2]-P[2] ) / (Q[1]-P[1]);

```

$$TVM := (P, Q) \mapsto \frac{Q_2 - P_2}{Q_1 - P_1} \quad (2.1)$$


Considérons de nouveau la fonction f définie par  $f(x) = \frac{1}{x+2}$  et illustrons l'interprétation graphique du

$TVM$   $\left[ \begin{matrix} -\frac{1}{2}, 3 \end{matrix} \right]$ .

```

> f:=x->1/(x+2):
P:=[-1/2,f(-1/2)]:
Q:=[3,f(3)]:
Courbe:=plot([x,f(x),x=-4..4]):
Points:=plot([P,Q],style=point,symbolsize=15,symbol=solidcircle,
color=navy):
Texte_P:=textplot([P[1]-0.25,P[2]+0.05,cat("P",convert(P,string))],
align=[below,left]):
Texte_Q:=textplot([Q[1],Q[2]+0.1,cat("Q",convert(Q,string))],align=
[below,right]):
Triangle:=plot([P,[P[1],Q[2]],Q],style=line,linestyle=2,color=black):
Éq_Sécante:=y=TVM(P,Q)*(x-P[1])+P[2]:
Droite:=plot([x,rhs(Éq_Sécante),x=-3..4],color=khaki):
display({Courbe,Droite,Triangle,Points,Texte_P,Texte_Q},view=[-1..4,
0..1.2]);

```



En déplaçant, **sur la courbe**, le point Q vers le point P, on obtient, pour chaque nouvelle position du point Q, une nouvelle sécante dont on pourrait calculer la pente. Pour ainsi déplacer le point Q, il faut lui donner des valeurs d'abscisses de plus en plus près de  $-\frac{1}{2}$ , soit la valeur d'abscisse du point P. (Notons que nous pouvons déplacer sur la courbe le point Q vers le point P car la fonction  $f$  ne possède aucune discontinuité entre P et Q)

Nous allons procéder comme suit:

–donnons au point Q la valeur de l'abscisse du point P plus une quantité  $h > 0$  que l'on fera tendre ultérieurement vers 0: soit donc  $Q\left(-\frac{1}{2} + h, f\left(-\frac{1}{2} + h\right)\right)$

–pour de telles accroissements  $h$ , considérons la suite:

$$2, 1, \frac{2}{3}, \frac{1}{2}, \frac{1}{3}, \frac{1}{4}, \frac{1}{5}, \frac{1}{10}, \frac{1}{20}, \frac{1}{30}, \frac{1}{40}, \frac{1}{50}, \frac{1}{100}, \frac{1}{1000}, \frac{1}{10000}, \frac{1}{100000}, \frac{1}{1000000}$$

Créons donc un point Q *variable* à l'aide la liste des accroissements  $h$  ci-dessus, laquelle liste sera nommée  $\Delta$  (Delta).

```
> Q:=[P[1]+h,f(P[1]+h)];
Delta:=[2, 1, 2/3, 1/2, 1/3, 1/4, 1/5, 1/10, 1/20, 1/30, 1/40, 1/50,
1/100, 1/1000,1/10000,1/100000,1/1000000];
```

$$Q := \left[ -\frac{1}{2} + h, \frac{1}{\frac{3}{2} + h} \right]$$

$$\Delta := \left[ 2, 1, \frac{2}{3}, \frac{1}{2}, \frac{1}{3}, \frac{1}{4}, \frac{1}{5}, \frac{1}{10}, \frac{1}{20}, \frac{1}{30}, \frac{1}{40}, \frac{1}{50}, \frac{1}{100}, \frac{1}{1000}, \frac{1}{10000}, \frac{1}{100000}, \frac{1}{1000000} \right] \quad (2.2)$$

Calculons maintenant tous les  $TVM_{\left[-\frac{1}{2}, -\frac{1}{2} + h\right]}$  associés à ces différents points Q. Pour ce faire, créons, avec la macro-commande `seq`, la séquence de ces valeurs qui sont les pentes des sécantes passant par les points P et Q, Q se rapprochant du point P.

```
> seq(TVM(P,Q),h in Delta);
- 4/21, - 4/15, - 4/13, - 1/3, - 4/11, - 8/21, - 20/51, - 5/12, - 40/93, - 10/23, - 80/183, - 25/57, - 200/453, - 2000/4503,
- 20000/45003, - 200000/450003, - 2000000/4500003, \quad (2.3)
```

$$h \rightarrow 0^+ \text{TVM}_{\left[-\frac{1}{2}, -\frac{1}{2} + h\right]} - \frac{20}{45} = -\frac{4}{9}$$

$$\text{TVM}_{\left[-\frac{1}{2}, -\frac{1}{2} + h\right]} \rightarrow -\frac{20}{45} = -\frac{4}{9} \text{ lorsque } h \rightarrow 0^+$$

Confirmons notre intuition en évaluant la limite à droite du  $\text{TVM}_{\left[-\frac{1}{2}, -\frac{1}{2} + h\right]}$  lorsque  $h \rightarrow 0^+$ .

```
> P:=[-1/2,'f'(-1/2)];
Q:=[-1/2+h,'f'(-1/2+h)];
Limit(TVM(P,Q),h=0,right);
normal(%);
```

$$P := \left[ -\frac{1}{2}, f\left(-\frac{1}{2}\right) \right]$$

$$Q := \left[ -\frac{1}{2} + h, f\left(-\frac{1}{2} + h\right) \right]$$

$$\lim_{h \rightarrow 0^+} \frac{\frac{1}{\frac{3}{2} + h} - \frac{2}{3}}{h}$$

$$\lim_{h \rightarrow 0^+} -\frac{4}{3(3 + 2h)} \tag{2.4}$$

Avec Maple, il n'aurait pas été nécessaire de simplifier explicitement (élimination du facteur  $h$ ) du TVM avant d'évaluer la limite, Maple le ferait à sa manière, mais ici, on a reproduit un développement *manu scriptus*.

```
> value((2.4));
```

$$-\frac{4}{9} \tag{2.5}$$

Lorsque la limite à droite existe, elle est appelée *taux de variation instantané à droite* en  $-\frac{1}{2}$  et est notée

$\text{TVI}_{\left[x=-\frac{1}{2}\right]}^+ = \lim_{h \rightarrow 0^+} \text{TVM}_{\left[-\frac{1}{2}, -\frac{1}{2} + h\right]}$ . Plus généralement on a la définition suivante:

### DÉFINITION

Le taux de variation instantané à droite en  $x = x_0$  d'une fonction  $f$  se définit de la façon suivante, lorsque la limite existe

$$\text{TVI}_{[x=x_0]}^+ = \lim_{h \rightarrow 0^+} \frac{f(x_0 + h) - f(x_0)}{h}$$

```
> (TVI^+)[[x=-1/2]]=Limit(TVM(P,Q),h=0,right);
`=Limit(normal(TVM(P,Q)),h=0,right);
value(%);
```

$$(TVI^+)_{\left[x=-\frac{1}{2}\right]} = \lim_{h \rightarrow 0^+} \frac{\frac{1}{\frac{3}{2} + h} - \frac{2}{3}}{h}$$

$$= \lim_{h \rightarrow 0^+} -\frac{4}{3(3 + 2h)}$$

$$= -\frac{4}{9} \quad (2.6)$$

Rappelons que les accents graves (voir [?backquotes](#)) sont des délimiteurs de création de noms (voir [?Names](#)). Bien que leurs usages pour la création de noms de variables avec des caractères accentués n'est pas souhaitable (c'est un peu lourd), dans ce cas-ci, leurs usages peuvent s'avérer appropriés pour une mise en forme éclairante des résultats des requêtes.

Vous savez que l'interprète ignore les espacements et les sauts de lignes dans l'énoncé d'une requête mais entre les deux délimiteurs de création de noms, les espaces sont interprétés comme autant de caractères blancs. Un caractère blanc est un caractère n'ayant aucun pixel noir. S'il n'y a aucun caractère entre ces deux délimiteurs, « `` », l'afficheur commencera l'affichage du résultat par un espace tout simplement.

Lorsque le TVI $_{[x=\frac{1}{2}]}$ <sup>+</sup> existe (c'est une limite), on dit que la *dérivée à droite en  $x = \frac{1}{2}$*  existe. Ici, cette dérivée vaut  $-\frac{4}{9}$ . La dérivée à droite est notée plus brièvement de la manière suivante.

$$f'_+\left(\frac{1}{2}\right) = -\frac{4}{9}$$

## DÉFINITION

La dérivée à droite d'une fonction  $f$  se définit de la façon suivante, lorsque la limite existe

$$f'_+(x) = \lim_{h \rightarrow 0^+} \frac{f(x+h) - f(x)}{h}$$


L'animation suivante montrera que si le point Q peut se rapprocher du point P par la droite, aussi près que l'on veut, les différentes sécantes tendront à se confondre avec la tangente qu'on pourrait tracer au point P du graphique de la fonction  $f$ .

```
> Éq_Sécante:=y=TVM(P,Q)*(x-P[1])+P[2]:
Delta:=[2, 1, 2/3, 1/2, 1/3, 1/4, 1/5, 1/10, 1/20, 1/30, 1/40, 1/50,
1/100, 1/1000,1/10000,1/100000,1/1000000]:
Texte_P:=seq(textplot([P[1],P[2],"P"],align={above,right}),n=1..nops
(Delta)):
Texte_Q:=seq(textplot([Q[1],Q[2],"Q"],align={above,right}),h in
Delta):
Sécantes:=seq(plot([[x,rhs(Éq_Sécante),x=-2..3],[x,f(x),x=-2..3]],
numpoints=100,color=[khaki,navy]),h in Delta):
Points:=seq(plot([P,Q],style=point,symbolsize=15,symbol=solidcircle,
color=navy),h in Delta):
Triangle:=seq(plot([P,[P[1],Q[2]],Q],style=line,linestyle=2,color=
black),h in Delta):
Accroissement:=seq(textplot([.15,.7,cat("Accroissement h = ",convert
(Delta[n],string))],
align={ABOVE,RIGHT},color=red),n=1..nops(Delta)):
Pente:=seq(textplot([.15,.6,cat("Pente m = ",convert(evalf(TVM(P,Q)),
string))],
align={ABOVE,RIGHT},color=red),h in Delta):
A:=display(Sécantes,insequence=true):
B:=display(Points,insequence=true):
```

```

C:=display(Triangle,insequence=true):
E:=display(Accroissement,insequence=true):
F:=display(Pente,insequence=true):
G:=display(Texte_P,insequence=true):
H:=display(Texte_Q,insequence=true):
display(A,B,C,E,F,G,H,view=[-1..2,0..1.2]);

```



Pour lancer l'animation, il faut cliquer d'abord sur le graphique pour le sélectionner : une bordure bleue atteste sa sélection. Un menu contextuel « Animation » est aussitôt sélectionné dans la barre contextuelle.

Dans une démarche similaire, nous allons créer une seconde animation pour illustrer le TVI à gauche du point P.

Puisque dans l'animation précédente les différentes valeurs d'accroissement  $h$  étaient positives, les différents points  $Q\left(-\frac{1}{2} + h, f\left(-\frac{1}{2} + h\right)\right)$  se retrouvaient tous à la droite du point P. Alors, pour réaliser une animation montrant, cette-fois, un point Q se rapprochant par la gauche, aussi près que l'on veut du point P, il suffit de considérer des valeurs d'accroissement  $h$  négatives.

Créons, à partir de la séquence  $\Delta$  et avec l'aide de la macro-commande `map` une séquence d'accroissements négatives. Donnez-lui le nom  $\Delta 2$  (Delta2). Attention, il nous faut laisser tomber la valeur d'accroissement  $h = 2$  de la liste  $\Delta$  car l'image  $f\left(-\frac{1}{2} + (-2)\right)$  ( $h = -2$ ) n'est pas sur la partie **continu** du graphique de  $f$  menant au point P. Alors `Delta[2..nops(Delta)]` précisera tous les éléments de la liste  $\Delta$  sauf le premier. La macro-commande `nops` donne le nombre d'opérandes composant un objet. Ici, le nombre d'éléments qui composent la liste  $\Delta$ .

```

> Delta1:=map(x->-x,Delta[2..nops(Delta)]);

$$\Delta 1 := \left[ -1, -\frac{2}{3}, -\frac{1}{2}, -\frac{1}{3}, -\frac{1}{4}, -\frac{1}{5}, -\frac{1}{10}, -\frac{1}{20}, -\frac{1}{30}, -\frac{1}{40}, -\frac{1}{50}, -\frac{1}{100}, -\frac{1}{1000}, -\frac{1}{10000}, \right. \\ \left. -\frac{1}{100000}, -\frac{1}{1000000} \right] \quad (2.7)$$


```


Copions-collons maintenant les requêtes de l'animation précédente et effectuons ensuite les modifications suivantes:

- substituons le nom  $\Delta$  par le nom  $\Delta 1$ . (Il y a huit substitutions)
- substituons P par Q et Q par P dans la structure graphique  $\textit{Triangle}$

- corrigeons la position verticale du texte *Accroissement* à 1.25 et celui de *Pente* à 1.05

- modifions à la fin l'option view par view=[-2..3,0..2.5].

```
> Éq_Sécante:=y=TVM(P,Q)*(x-P[1])+P[2]:
Texte_P:=seq(textplot([P[1],P[2],"P"],align={above,right}),n=1..nops
(Delta)):
Texte_Q:=seq(textplot([Q[1],Q[2],"Q"],align={above,right}),h in
Delta):
Sécantes:=seq(plot([[x,rhs(Éq_Sécante),x=-2..3],[x,f(x),x=-2..3]],
numpoints=100,color=[khaki,navy]),h in Delta):
Points:=seq(plot([P,Q],style=point,symbolsize=15,symbol=solidcircle,
color=navy),h in Delta):
Triangle:=seq(plot([Q,[Q[1],P[2]],P],style=line,linestyle=2,color=
black),h in Delta):
Accroissement:=seq(textplot([.15,1.25,cat("Accroissement h = ",
convert(Delta[n],string))],
align={ABOVE,RIGHT},color=red),n=1..nops(Delta)):
Pente:=seq(textplot([.15,1.05,cat("Pente m = ",convert(evalf(TVM(P,Q)
),string))],
align={ABOVE,RIGHT},color=red),h in Delta):
A:=display(Sécantes,insequence=true):
B:=display(Points,insequence=true):
C:=display(Triangle,insequence=true):
E:=display(Accroissement,insequence=true):
F:=display(Pente,insequence=true):
G:=display(Texte_P,insequence=true):
H:=display(Texte_Q,insequence=true):
display(A,B,C,E,F,G,H,view=[-2..3.5,0..2.5]);
```



Obtenons, ici aussi, tous les  $TVM\left[-\frac{1}{2}+h,-\frac{1}{2}\right]$  associés à ces différents points Q.

```
> seq(TVM(P,Q),h=Delta);
-4/3, -4/5, -2/3, -4/7, -8/15, -20/39, -10/21, -40/87, -5/11, -80/177, -50/111, -200/447, -2000/4497, -20000/44997, (2.8)
```

$$\left[ -\frac{200000}{449997}, -\frac{2000000}{4499997} \right]$$

Nous observons ici aussi que plus l'accroissement  $h \rightarrow 0^-$  plus le  $TVM_{\left[-\frac{1}{2} + h, -\frac{1}{2}\right]}$  se rapproche de  $-\frac{20}{45} = -\frac{4}{9}$

$$TVM_{\left[-\frac{1}{2} + h, -\frac{1}{2}\right]} \rightarrow -\frac{20}{45} = -\frac{4}{9} \text{ lorsque } h \rightarrow 0^-$$

Lorsque la limite à gauche existe, elle est appelée *taux de variation instantané à gauche* en  $-\frac{1}{2}$  et est notée

$$TVI_{\left[x=-\frac{1}{2}\right]}^- = \lim_{h \rightarrow 0^+} \frac{f\left(-\frac{1}{2} + h\right) - f\left(-\frac{1}{2}\right)}{h}. \text{ Plus généralement on a la définition suivante}$$

### DÉFINITION

Le taux de variation instantané à gauche en  $x = x_0$  d'une fonction  $f$  se définit de la façon suivante, lorsque la limite existe

$$TVI_{\left[x=x_0\right]}^- = \lim_{h \rightarrow 0^-} \frac{f(x_0 + h) - f(x_0)}{h}$$

```
> (TVI^-) [[x=-1/2]] = Limit (TVM(P,Q), h=0, left);
    ^ = normal(rhs(%));
    value(%);
```

$$\begin{aligned} (TVI^-)_{\left[x=-\frac{1}{2}\right]} &= \lim_{h \rightarrow 0^-} \frac{\frac{1}{\frac{3}{2} + h} - \frac{2}{3}}{h} \\ &= \lim_{h \rightarrow 0^-} -\frac{4}{3(3 + 2h)} \\ &= -\frac{4}{9} \end{aligned} \tag{2.9}$$

Lorsque le  $TVI_{\left[x=-\frac{1}{2}\right]}^-$  existe (c'est une limite), on dit que la *dérivée à gauche* en  $x = \frac{1}{2}$  existe. Ici, cette dérivée vaut  $-\frac{4}{9}$ . Cette dérivée à gauche est notée plus brièvement de la manière suivante.

$$f'_+ \left( -\frac{1}{2} \right) = -\frac{4}{9}$$

### DÉFINITION

La dérivée à gauche d'une fonction  $f$  se définit de la façon suivante, lorsque la limite existe

$$f'_-(x) = \lim_{h \rightarrow 0^-} \frac{f(x+h) - f(x)}{h}$$

Puisque la dérivée à gauche existe, que la dérivée à droite existe et qu'elles sont égales, on dit que la fonction  $f$  est dérivable en  $x = -\frac{1}{2}$ .

$$f' \left( -\frac{1}{2} \right) = -\frac{4}{9}$$

**REMARQUE IMPORTANTE:** Pour définir la dérivée en  $x = x_0$  il faut absolument que l'on puisse approcher  $(x_0, f(x_0))$  sur la courbe par la droite et pas la gauche. Il faut donc que la fonction  $f$  soit continue dans tout

voisinage de  $x_0$ :  $f$  doit être continue à droite de  $x_0$  et à gauche de  $x_0$ .

C'est une condition nécessaire.

- Si la fonction  $f$  n'est pas continue en  $x_0$  alors la fonction  $f$  n'est pas dérivable en  $x_0$ .

Mais ce n'est pas suffisant.

- Si la fonction  $f$  est continue en  $x_0$  alors la fonction  $f$  n'est pas nécessairement dérivable en  $x_0$ .

En effet,

- d'une part, si l'une des deux dérivées directionnelles n'existent pas en  $x_0$ , a fortiori la dérivée n'existe pas en  $x_0$ .
- d'autre part, même si la dérivée à gauche en  $x_0$  et la dérivée à droite en  $x_0$  existent toutes les deux, elles peuvent ne pas être égales (ce sont des limites).

Un exemple classique est la fonction  $f$  définie par  $f(x) = |x|$ . En  $x_0 = 0$ , la fonction  $f$  est continue mais pas dérivable en  $x_0 = 0$ .

Analysons la continuité de  $f$  en  $x_0 = 0$ .

```
> f:=x->abs(x);
```

$$f := x \mapsto |x| \tag{2.10}$$

```
> Limit(f(x),x=0,left);
assume(x<0);
``=Limit(f(x),x=0,left);
``=value(rhs(%));
x:='x':
```

$$\lim_{x \rightarrow 0^-} |x| = \lim_{x \rightarrow 0^-} (-x) = 0 \tag{2.11}$$

```
> Limit(f(x),x=0,right);
assume(x>0);
``=Limit(f(x),x=0,right);
``=value(rhs(%));
x:='x':
```

$$\lim_{x \rightarrow 0^+} |x| = \lim_{x \rightarrow 0^+} x = 0 \tag{2.12}$$

Puisque les deux limites directionnelles sont égales,  $f$  est continue en  $x_0 = 0$ .

Analysons ensuite la dérivabilité de  $f$  en  $x_0 = 0$ .

```
> P:=[0,'f'(0)];
Q:=[0+h,'f'(0+h)];
(TVI^`-`)[[x=0]]=Limit(TVM(P,Q),h=0,left);
assume(h<0);
``=rhs(%);
``=value(rhs(%));
```



```
h:='h':
```

$$\begin{aligned} (TVI^{\prime-})_{[x=0]} &= \lim_{h \rightarrow 0^-} \frac{|h|}{h} \\ &= \lim_{h \rightarrow 0^-} (-1) \\ &= -1 \end{aligned}$$

(2.13)

```
> (TVI^{\prime+})[[x=0]]=Limit(TVM(P,Q),h=0,right);  
assume(h>0);  
``=rhs(%);  
``=value(rhs(%));  
h:='h':
```

$$\begin{aligned} (TVI^{\prime+})_{[x=0]} &= \lim_{h \rightarrow 0^+} \frac{|h|}{h} \\ &= \lim_{h \rightarrow 0^+} 1 \\ &= 1 \end{aligned}$$

(2.14)

Même si les dérivées à gauche et à droite de  $f$  en  $x_0 = 0$  existent, elles ne sont pas égales, la fonction  $f$  n'est donc pas dérivable en  $x_0 = 0$ .